

Specification of the signature-free *Sailfish* consensus algorithm at a high level of abstraction.

We use a number of abstractions and simplifying assumptions in order to expose the high-level principles of the algorithm clearly and in order to make model-checking of interesting configurations tractable:

- 1) Nodes read and write a global *DAG*. When a node transitions into a new round, it is provided with an arbitrary quorum of vertices from the previous round (except that, after *GST*, some additional assumptions apply).
- 2) We do not model timeouts. Instead, we assume that, every round r after *GST*, each correct node votes for the previous leader.
- 3) We model *Byzantine* nodes explicitly by assigning them an algorithm too. This algorithm should allow the worst attacks possible, but, while the author thinks this is true, there is no formal guarantee that this is the case. A more realistic model would allow *Byzantine* nodes to send completely arbitrary messages at any time, but this would make model-checking with *TLC* too hard.
- 4) We do not explicitly model committing based on $2f + 1$ first *RBC* messages.
- 5) There are no weak edges.

EXTENDS *Digraph*, *Integers*, *FiniteSets*, *Sequences*

CONSTANTS

- N The set of all nodes
- , F The set of *Byzantine* nodes
- , R The set of rounds
- , $IsQuorum(-)$ Whether a set is a quorum (*i.e.* cardinality $\geq n-f$)
- , $IsBlocking(-)$ Whether a set is a blocking set (*i.e.* cardinality $\geq f + 1$)
- , $Leader(-)$ operator mapping each round to its leader
- , GST the first round in which the system is synchronous

ASSUME $\exists n \in R : R = 1 \dots n$ useful rounds start at 1

For our purpose of checking safety and liveness of *Sailfish*, we do not need to model blocks of transactions. Instead, *DAG* vertices just consist of a node and a round.

$V \triangleq N \times R$ the set of possible *DAG* vertices
 $Node(v) \triangleq v[1]$
 $Round(v) \triangleq v[2]$

Next we define how we order *DAG* vertices when we commit a leader vertice

$LeaderVertice(r) \triangleq \langle Leader(r), r \rangle$

RECURSIVE $OrderSet(-)$ arbitrarily order the members of the set S

$OrderSet(S) \triangleq$ IF $S = \{\}$ THEN $\langle \rangle$ ELSE
 LET $e \triangleq$ CHOOSE $e \in S : \text{TRUE}$
 IN $Append(OrderSet(S \setminus \{e\}), e)$

NOTE: CHOOSE is deterministic in TLA+,
i.e. CHOOSE $e \in S : \text{TRUE}$ is always the same e if S is the same

```

RECURSIVE CollectLeaders(−, −, −)
CollectLeaders(vs, r, dag)  $\triangleq$  IF vs = {} THEN ⟨⟩ ELSE
  LET children  $\triangleq$  UNION {Children(v, dag) : v ∈ vs}
  IN IF LeaderVertice(r) ∈ vs
    THEN Append(
      CollectLeaders(Children(LeaderVertice(r), dag), r − 1, dag),
      LeaderVertice(r))
    ELSE CollectLeaders(children, r − 1, dag)

RECURSIVE OrderVertices(−, −)
OrderVertices(dag, leaderVertices)  $\triangleq$ 
  IF leaderVertices = ⟨⟩ THEN ⟨⟩ ELSE
  LET l  $\triangleq$  Head(leaderVertices)
    toOrder  $\triangleq$  Descendants({l}, dag)
    prefix  $\triangleq$  OrderSet(toOrder)
    remainingVertices  $\triangleq$  Vertices(dag) \ (toOrder ∪ {l})
    remainingEdges  $\triangleq$  {e ∈ Edges(dag) : {e[1], e[2]} ⊆ remainingVertices}
    remainingDAG  $\triangleq$  ⟨remainingVertices, remainingEdges⟩
  IN prefix ∘ ⟨l⟩ ∘ OrderVertices(remainingDAG, Tail(leaderVertices))

CommitLeader(v, dag)  $\triangleq$ 
  LET leaderVertices  $\triangleq$  CollectLeaders({v}, Round(v), dag)
  IN OrderVertices(dag, leaderVertices)

```

Now we specify the algorithm in the *PlusCal* language.

```

--algorithm Sailfish{
  variables
    vs = {},    the vertices of the DAG
    es = {};    the edges of the DAG
  define {
    dag  $\triangleq$  ⟨vs, es⟩
    NoVoteQuorum(r, delivered)  $\triangleq$ 
      LET NoVote  $\triangleq$  {v ∈ delivered : LeaderVertice(r − 1) ∉ Children(v, dag)}
      IN IsQuorum({Node(v) : v ∈ NoVote})
    ValidLeaderVertice(lv, delivered)  $\triangleq$  LET r  $\triangleq$  Round(lv) IN
      ∨ r = 1
      ∨ LeaderVertice(r − 1) ∈ Children(lv, dag)
      ∨ NoVoteQuorum(r, delivered)
  }
  process ( correctNode ∈ N \ F )
    variables
      round = 0,    current round; 0 means the node has not started execution
      log = ⟨⟩;    delivered log

```

```

{
l0:  while ( TRUE ) { keep starting new rounds
      round := round + 1 ;
      with ( newV = ⟨self, round⟩ ) {
        if ( round = 1 )
          vs := vs ∪ {newV}
        else with ( delivered ∈ SUBSET {v ∈ vs : Round(v) = round - 1} ) {
          await IsQuorum({Node(v) : v ∈ delivered});
          await LeaderVertice(round - 1) ∈ delivered
            ⇒ ValidLeaderVertice(LeaderVertice(round - 1), delivered);
          if ( Leader(round) = self )
            await ValidLeaderVertice(newV, delivered);
          vs := vs ∪ {newV};
          es := es ∪ {⟨newV, pv⟩ : pv ∈ delivered};
          commit if there is a quorum of votes for the leader of r - 2:
          if ( round > 1 )
            with ( votesForLeader = {pv ∈ delivered : ⟨pv, LeaderVertice(round - 2)⟩ ∈ es} )
              if ( IsBlocking({Node(pv) : pv ∈ votesForLeader}) )
                log := CommitLeader(LeaderVertice(round - 2), dag)
            }
          }
        }
      }
}

```

Next comes our model of *Byzantine* nodes. Because the real protocol disseminates *DAG* vertices using reliable broadcast, *Byzantine* nodes cannot equivocate and cannot deviate much from the protocol (lest their messages be ignored). Also note that creating a round- r vertex commutes to the left of actions of rounds greater than r and to the right of actions of rounds smaller than R , so we can, without loss of generality, schedule *Byzantine* nodes in the same “round-by-round” manner as other nodes.

```

process ( byzantineNode ∈ F )
  variables round_ = 0 ;
  {
l0:  while ( TRUE ) {
      round_ := round_ + 1 ;
      maybe add a vertices to the DAG:
      either with ( newV = ⟨self, round_⟩ ) {
        if ( round_ = 1 )
          vs := vs ∪ {newV}
        else
          with ( delivered ∈ SUBSET {v ∈ vs : Round(v) = round_ - 1} ) {
            await IsQuorum({Node(v) : v ∈ delivered});
            vs := vs ∪ {newV};
            es := es ∪ {⟨newV, pv⟩ : pv ∈ delivered}
          }
      }
    } or skip ;
  }

```

$$\} \\ \} \\ \}$$

Next we define the safety and liveness properties

$$\begin{aligned} \text{Compatible}(s1, s2) &\triangleq \text{whether the sequence } s1 \text{ is a prefix of the sequence } s2, \text{ or vice versa} \\ \text{LET } \text{Min}(n1, n2) &\triangleq \text{IF } n1 \geq n2 \text{ THEN } n2 \text{ ELSE } n1 \text{ IN} \\ &\quad \forall i \in 1 \dots \text{Min}(\text{Len}(s1), \text{Len}(s2)) : s1[i] = s2[i] \end{aligned}$$

$$\text{Agreement} \triangleq \forall n1, n2 \in N \setminus F : \text{Compatible}(\log[n1], \log[n2])$$

$$\begin{aligned} \text{Liveness} &\triangleq \forall r \in R : r \geq \text{GST} \wedge \text{Leader}(r) \notin F \Rightarrow \\ &\quad \forall n \in N \setminus F : \text{round}[n] \geq r + 2 \Rightarrow \\ &\quad \exists i \in \text{DOMAIN } \log[n] : \log[n][i] = \text{LeaderVertice}(r) \end{aligned}$$

Finally we make a few auxiliary definitions used for model-checking with *TLC*

The round of a node, whether *Byzantine* or not

$$\text{Round}_-(n) \triangleq \text{IF } n \in F \text{ THEN } \text{round}_-[n] \text{ ELSE } \text{round}[n]$$

Basic typing invariant:

$$\begin{aligned} \text{TypeOK} &\triangleq \\ &\quad \wedge \forall v \in \text{vs} : \text{Node}(v) \in N \wedge \text{Round}(v) \in \text{Nat} \wedge \text{Round}(v) > 0 \\ &\quad \wedge \forall e \in \text{es} : \\ &\quad \quad \wedge e = \langle e[1], e[2] \rangle \\ &\quad \quad \wedge \{e[1], e[2]\} \subseteq \text{vs} \\ &\quad \quad \wedge \text{Round}(e[1]) > \text{Round}(e[2]) \\ &\quad \wedge \forall n \in N : \text{Round}_-(n) \in \text{Nat} \end{aligned}$$

Synchrony assumption: for each round r from *GST* onwards, if the leader of r is correct then every correct node votes for the round- r leader vertices in round $r + 1$

$$\begin{aligned} \text{Synchrony} &\triangleq \forall r \in R : r \geq \text{GST} \wedge \text{Leader}(r) \notin F \Rightarrow \\ &\quad \forall n \in N \setminus F : \text{LET } v \triangleq \langle n, r + 1 \rangle \text{ IN} \\ &\quad \quad v \in \text{vs} \Rightarrow \text{LeaderVertice}(r) \in \text{Children}(v, \text{dag}) \end{aligned}$$

Sequentialization constraints, which enforce a particular ordering of the actions. Because of how actions commute, the set of reachable states remains unchanged. Essentially, we schedule all nodes “round-by-round” and in lock-steps, with the leader last. This speeds up model-checking a lot.

Note that we must always schedule the leader last because, because of its relying on other nodes’s vertices, its action does not commute to the left of the actions of other nodes.

An arbitrary ordering of the nodes, with the round leader last:

$$\begin{aligned} \text{NodeSeqLeaderLast}(r) &\triangleq \text{CHOOSE } s \in [1 \dots \text{Cardinality}(N) \rightarrow N] : \\ &\quad \wedge s[\text{Cardinality}(N)] = \text{Leader}(r) \\ &\quad \wedge \forall i, j \in 1 \dots \text{Cardinality}(N) : i \neq j \Rightarrow s[i] \neq s[j] \\ \text{NodeIndexLeaderLast}(n, r) &\triangleq \text{CHOOSE } i \in 1 \dots \text{Cardinality}(N) : \text{NodeSeqLeaderLast}(r)[i] = n \end{aligned}$$

$$\begin{aligned}
SeqConstraints(n) &\triangleq \\
&\text{wait for all nodes to be at least in the round:} \\
&\wedge \forall n2 \in N : Round_ (n2) \geq Round_ (n) \\
&\text{wait for all nodes with lower index to leave the round (leader index is always last):} \\
&\wedge \forall n2 \in N : NodeIndexLeaderLast(n2, Round_ (n)) < NodeIndexLeaderLast(n, Round_ (n)) \\
&\Rightarrow Round_ (n2) > Round_ (n)
\end{aligned}$$

We add the sequentialization constraints and the synchrony assumption to the specification

$$\begin{aligned}
SeqNext &\triangleq (\exists self \in N \setminus F : SeqConstraints(self) \wedge correctNode(self) \wedge Synchrony') \\
&\quad \vee (\exists self \in F : SeqConstraints(self) \wedge byzantineNode(self)) \\
SeqSpec &\triangleq Init \wedge \Box [SeqNext]_{vars}
\end{aligned}$$

Next we define a constraint to stop the model-checker.

$$\begin{aligned}
Max(S) &\triangleq \text{CHOOSE } x \in S : \forall y \in S : y \leq x \\
StateConstraint &\triangleq \forall n \in N : Round_ (n) \in 0 \dots Max(R)
\end{aligned}$$

Finally, we give some properties we expect to be violated (useful to get the model-checker to print interesting executions).

$$\begin{aligned}
Falsy1 &\triangleq \neg(\\
&\quad \forall n \in N : Round_ (n) = Max(R) \\
&)
\end{aligned}$$

$$\begin{aligned}
Falsy2 &\triangleq \neg(\\
&\quad \exists n \in N \setminus F : Len(log[n]) > 1 \\
&)
\end{aligned}$$
